include(../../../mkspecs/qmake/app.pri)
include(../../../mkspecs/qmake/version.pri)
include(../../../mkspecs/qmake/install_lib_qt.pri)

QT              += core gui widgets
TARGET           = app-qt$${DEBUG_POSTFIX}

TEMPLATE         = app
DEFINES         += QT_DEPRECATED_WARNINGS

SOURCES         += \
                   main.cpp \
                   mainwindow.cpp

HEADERS         += \
                   mainwindow.h

FORMS           += \
                   mainwindow.ui

win32::RC_FILE   = app-qt.rc

INCLUDEPATH     += \
                   $${CYG_INC_PATH}/mod1-shared-ab \
                   $${CYG_INC_PATH}/mod2-shared-ab

LIBS            += -lmod1-shared-ab$${DEBUG_POSTFIX}
LIBS            += -lmod2-shared-ab$${DEBUG_POSTFIX}

# Install app run scripts
run.files       += qt.conf
win32:run.files += app-qt$${DEBUG_POSTFIX}.bat
unix:run.files  += app-qt$${DEBUG_POSTFIX}.sh
INSTALLS        += run

# Install configure files
config.files    += app-qt.json
INSTALLS        += config

# Install linux package desktop and icons files
unix {
desktop.files   += share/applications/app-qt$${DEBUG_POSTFIX}.desktop
icons.files     += share/icons
INSTALLS        += desktop icons
}
